
For this composition, I used Keykit, rather than Csound, because I was most comfortable implementing the algorithms that I wanted to use in Keykit.  Unlike Csound’s flexible, and sometimes confusing, ways of implementing instruments and scores, Keykit is more like a structured programming language, with similar methods and syntax to Python and Perl.  During my experiences with Keykit and Csound, I found that I enjoyed algorithmically generating notes, and then fine-tuning them, rather than having complete control over the structure and composition of the pitches.  Because I have never had any musical training, I felt very insecure picking notes and beats arbitrarily, and to a certain extent, even working with them in general.  However, after establishing a structure I was comfortable with—even though it changed many times—how I wanted to implement my program to generate a song came through smoothly.


The basis of the code I used came from a prior homework, which used implementations of a pentatonic function, and a context-free grammar ruleset, which chooses among multiple rules randomly.  The pentatonic function was simple in its design.  The function would take in a phrase—a group of notes—and compare each note to its closest neighbor within a certain scale, in this case, the c major scale.  If it was the same note, it would keep it as-is, but if it was not, it used a Keykit function that returns the closest note, by passing the note in question as well as a list of acceptable notes.  As this function cycled through the original phrase passed, it would create another phrase that it put the new notes, unchanged or not, inside.  After all the notes are evaluated, it returns that new phrase.  For simplicity’s sake, I continued to use the c major scale for my notes in the final composition, even though I could have easily replaced the list of notes in the c major scale with another set of notes from another scale.


The context-free grammar functions are the functions I used most often in my code.  I wrote a rule for each note, and each rule had two or three options of rules that it could enforce, which are randomly chosen.  The rules consist of functions named for which situation they would be used in—in this case, each rule is named after a note.  In this version, the implementation of the ruleset only considered the note in question, and not any information about duration, or any other notes in the phrase.  Without any information about the phrase as a whole, this implementation of the grammatical structure is considered context-free.  The rules are enforced by a function called propagate, which takes in a phrase of notes, and returns a completely different phrase, built up from the notes in the original phrase.


To create a rhythm for this group of notes I could generate, I wrote a function for each rhythm that would take the phrase of notes, and set each one to a specific duration and place in time within that phrase.  This function was first implemented in my previous attempt at a longer composition, but as it turns out, was assigning the duration and time values incorrectly.  This was one of the bigger bugs in the later code that made the previous composition have a unique sound.  However, because of this bug in the original code, I had a hard time working with the rhythms that I made for this composition.  It was about one-third of the way through the process of creating the sections when I discovered my code was not working correctly. It changed the way my composition sounded drastically.  This meant I had to go back and restructure my rhythm and timing map, which, ultimately, was a good step to take.


I wanted a similar rhythm to the map that I used for my previous composition, so I started out with a section of that original rhythm, and built off that.  Most of the rhythmic structure is based around two or three shorter notes, with a longer note wrapping up the measure.  I still was not sure how I wanted the whole piece to be structured at this point.  Figuring out the basic rhythm ended up being a trial and error process, running the rhythm function with different generated phrases, and making notes on how the notes, and sometimes the absence of notes, worked with the general rhythm.


I was inspired by Tiffany Chong’s snippet of her final composition that she played in class to have two different instruments playing the same phrase, though at different times.  I was working on my composition in stages, so I started with the beginning piece.  I wanted it to be a setup for the rest of the piece, and, while not necessarily start with dissonance between the two instruments, I wanted there to be a very subtle clash at some parts, but still go together as a whole.  With all the other compositions I had done before, I had not known how to change the midi instrument on a channel, so I had been working with the default piano instrument.  Because of this, I tried to stay as far away from the piano instrument as I could.  I experimented with putting two different instruments together in exact parallel, to see if I could get a sound that I liked.  I ended up giving up on that avenue after listening to the instruments being played in class, while I was presenting my final project in progress.


At this point in the project, I knew the basic structure that I wanted.  I envisioned a beginning piece with two instruments playing at different times, but with the same rhythm, abruptly cutting off into a solo of some sort from one of those instruments.  The climax would be the two instruments once again, though I was not sure how I wanted that to culminate.  I also knew that I wanted the resolution of the piece to be the two instruments playing in unison.  Because of this, I needed two instruments that sounded good together in unison.  Not knowing where to begin, I decided to listen to some other midi files that I had.


One game that I had been playing recently, Legend of Zelda, had a certain song called Zelda’s Lullaby that was played frequently in the game.  I liked the rhythm, and the feel of the song, so I tried to see if I could take the midi apart, and listen to the separate instruments.  I got two inspirations from doing this, though I could not deduce how I could reproduce the rhythm of the original song.  The first idea I got was to use the ocarina instrument for one of the instruments.  It had the kind of rich voice I was looking for, and, even though I had been listening to piano music from Keykit for half the semester, I decided to use the piano instrument as the second instrument, because of the way they sounded together.  The second inspiration I had was the use of an overarching instrumental section, with the notes holding across several of the notes of the other instruments.  I started it off as a string instrument, and kept it like that, because it seemed to fit pretty well with the other instruments and the theme I was looking for.  One problem that I immediately ran into was having this part of the piece fit the length of the other pieces.  My implementation of this section extended the lengths of the base rhythm I was using in the piece, and stretched the time between the notes as well.  This made it necessary to either cut the generated phrase short manually, or only generate as many notes as it would take to play the duration of the main composition.  Since this problem had come up once before, I decided to write a function that would cut a phrase of notes down to a specified number.  Unfortunately, by doing this, I introduced another time-consuming bug—the phrase it would return had no rhythm information.  Fortunately, I got around that by not altering the rhythms of each phrase I was using until after they were cut to the length I needed.


For the undertones of the middle part of the piece, I wanted an instrument that would play a similar melody, with the duration stretched like the string section that spanned the length of the composition.  To do this, I made a completely new rhythm, basing it off the rhythm that I was using.  The main difference was I combined some parts of the rhythm to make the notes hold instead of transitioning to other notes.  I ended up using the viola instrument, which sounded good on my computer, but when brought to the synthesizer, I was surprised how different it sounded.  After working out how I wanted the underlying rhythm to be laid out, I was at a loss as to how to make the middle part of the song a buildup to the climax.  I knew that, for the climax, I wanted all of the instruments to be playing, and the string instrument that spanned the entire composition to be playing frequently.  However, I wanted some breakup between the beginning introductory phrase, and the climax, which involved the main element for the beginning.


One thing that I had wanted to experiment with was the idea of improvisation.  After the viola instrument was introduced in the composition, I inserted a section with the ocarina instrument that used one of the ideas I came up with.  I still wanted to use the rhythm that I started out with, but I wanted it to sound distinctly different from the rest.  To accomplish this, I generated multiple short phrases, using the same grammar I used for the intro and the outtro.  I then propagated a few of them, to change the grammatical structure of the notes, and set them to the same rhythm, but in different parts.  After that, I concatenated the phrases, and it became the middle part of the middle phrase.


The climax was made up of the same beginning phrases, except now the viola was playing in the background, and the strings in the foreground.  I turned up the volume for the piano so it could be the dominant instrument, and because of that, it still held the dominant timing of the rhythm.  It was too hectic at first, and required the pause between the “improvised” section and the climax section to be manually played with until it sounded reasonable.  In contrast to most of the other parts of the piece, I didn’t try to tweak the phrases, and relied on the rhythms to resolve their notes into pleasant melodies.


For the outtro, I wanted to resolve the rhythm conflict I had been setting up during the entire composition. To do this, I decided to shorten the difference of time between the two instruments playing, until they played in parallel.  I also appended the first note the rhythm started at to end the composition—it seemed less abrupt than just stopping when the rhythm stopped.  Looking back, I think the two instruments should have resolved faster.  By the time they do resolve, it feels like the audience has already forgotten about the climax.  Overall, the composition felt almost like a lullaby, which was one of the ideas that got me inspired to make many of the phrases anyway.  Because of that, I can finally say that I actually enjoy listening to one of my compositions of music.

Appendix: KeyKit Code

function play()

{


light = overture()


main_instrument = mainPhrase()


comp = light | main_instrument


return( comp )

}

function mainPhrase()

{


# instruments set on channels


piano = progchange(1,2) 


ocarina = progchange(80,3)


underRhythm = ''


underRhythm = A() + E() + B()


while (underRhythm.length < 360*96 ) {



underRhythm = propagate(underRhythm)


}


underRhythm.chan = 3


piano_echo = underRhythm


piano_echo.chan = 2


# ########## beginning ########## #


# cut the phrases short


begin_piano = rhythm1(cut_phrase( piano_echo, 21*4 )) | piano


begin_ocarina = rhythm1(cut_phrase( underRhythm, (21*4)-9 )) | ocarina


# sets the ocarina back a bit


begin_ocarina.time += 768


beginning = strip(begin_piano|begin_ocarina)


# ###### middle ###### #


middle = rhythm2(cut_phrase(firstPhrase(), 21*6))


# the 'improvisation' of the ocarina at the beginning


impro = ''


impro = improvise() 


# after the impro, we bring in the beginning again, all together!


mid_ocarina = begin_ocarina


mid_ocarina.time -= 96 


mid_piano = begin_piano


mid_piano.vol += 10 # turned up the volume on the piano to give it more of an effect


after_impro = ( mid_piano | mid_ocarina )


after_impro.time += 512


impro += after_impro


impro.time += 1712


# put it over the viola


middle = middle | impro


# ########### end ########## #


# once again, we use the beginning phrase


end_piano = rhythm1(cut_phrase( piano_echo, 21 )) | piano


end_ocarina = rhythm1(cut_phrase( underRhythm, 21)) | ocarina


# overlap shrinks until they are in unison


overlap = end_ocarina


end = ''


overlap.time += 768 # starts off big (the overlap)


end += end_piano | overlap


overlap.time -= 256


end += end_piano | overlap


overlap.time -= 256


end += end_piano | overlap


overlap.time -= 80 # and continues downward until the ocarina and piano match


end += end_piano | overlap


overlap.time -= 176 # 


end += end_piano | overlap


# stick a little end piece to wrap it up


final = A()


final = rhythm1(final)


final.chan = 2 # piano ends off?


final |= piano


end += final


return(beginning+middle+end)

}

function improvise() {


# instruments set on channels


ocarina = progchange(80,7)


impro = ''


impro = A() + E() + B()


one = strip(rhythm1(impro))


impro = A() + E() + B()


two = strip(rhythm1(impro))


impro = propagate(impro)


three = strip(rhythm1(impro))


impro = A() + E() + B()


four = strip(rhythm1(impro))


while (impro.length < 2048*3 ) {



impro += one+two+three+four



one = rhythm1(propagate(one))



two = rhythm1(propagate(two))



three = rhythm1(propagate(three))



four = rhythm1(propagate(four))


}


impro.chan = 7


return(impro|ocarina)

}

function overture () {


# sort of like the bass I guess


instr = progchange(49, 8)


phr = A() + E() + B()


while(phr.length < 400*96)



phr = propagate(phr)


phr.chan = 8


newphr = ''


rhythm = [0=1, 1=128, 2=128, 3=128, 4=256, 5=96, 6=352, 7=96, 8=96, 9=256, 10=128, 11=256, 12=128, 13=96, 14=128, 15=96, 16=96, 17=256, 18=128, 19=128, 20=352, 21=256]


time = [0=1, 1=96, 2=96, 3=96, 4=128, 5=48, 6=176, 7=48, 8=48, 9=128, 10=96, 11=128, 12=96, 13=48, 14=96, 15=48, 16=48, 17=128, 18=96, 19=96, 20=176, 21=128]


current_beat = 1


current_time = 0


newnote = ''


counter =  time[current_time]


for ( note in phr )


{



if ( current_beat > 21 )




current_beat = 1



newnote = note



newnote.dur = rhythm[current_beat] * 3 



newnote.length = newphr.length + time[current_time]



newnote.time = counter



if ( current_time < 22 )




newphr |= newnote



current_beat += 1



current_time += 1



counter += time[current_time] * 17


}


newphr.vol=40


return(newphr|instr)

}

function firstPhrase()

{


viola = progchange(42,5)


underRhythm = ''


underRhythm = G() + E() + G() + G()


while (underRhythm.length < 150*96 ) {



underRhythm = propagate(underRhythm)


}


underRhythm.chan = 5


underRhythm.vol = 50


return((underRhythm|viola))

}

function rhythm2( phr )

{


endphr = ''


newnote = ''


rhythm = [0=1, 1=128, 2=80, 3=256, 4=96, 5=96, 6=96, 7=128, 8=128, 9=256, 10=128, 11=160, 12=256, 13=96, 14=96, 15=96, 16=256, 17=128, 18=128, 19=256, 20=96, 21=256]


current_beat = 1


current_time = 0


for ( note in phr )


{



if ( current_beat > 21 )




current_beat = 1



newnote = note



newnote.dur = rhythm[current_beat] 



newnote.length = endphr.length + rhythm[current_beat]



newnote.time = current_time



endphr |= newnote



current_time += rhythm[current_beat]



current_beat += 1


}


endphr = pent(endphr)


return(endphr)

}

function rhythm1( phr )

{


endphr = ''


newnote = ''


rhythm = [0=1, 1=96, 2=96, 3=96, 4=128, 5=48, 6=176, 7=48, 8=48, 9=128, 10=96, 11=128, 12=96, 13=48, 14=96, 15=48, 16=48, 17=128, 18=96, 19=96, 20=176, 21=128]


current_beat = 1


current_time = 0


for ( note in phr )


{



if ( current_beat > 21 )




current_beat = 1



newnote = note



newnote.dur = rhythm[current_beat] 



newnote.length = endphr.length + rhythm[current_beat]



newnote.time = current_time



endphr |= newnote



current_time += rhythm[current_beat]



current_beat += 1


}


endphr = pent(endphr)


return(endphr)

}

function propagate( ph )

{


endph = ''


# creates a grammar according to a set of notes


for ( note in ph )


{



if ( note.pitch == 'a'.pitch )




endph += A()



if ( note.pitch == 'b'.pitch )




endph += B()



if ( note.pitch == 'c'.pitch )




endph += C()



if ( note.pitch == 'd'.pitch )




endph += D()



if ( note.pitch == 'e'.pitch )




endph += E()



if ( note.pitch == 'f'.pitch )




endph += F()



if ( note.pitch == 'g'.pitch )




endph += G()



if( note.pitch == 0 )




endph += 'r'


}


return(endph)

}

function C()

{


# C -> ErF | AG | G


ph = ''


x = rand(1,3)


if ( x == 1 ) 



ph += 'e,r,f'


if ( x == 2 ) 



ph += 'a,g'


if ( x == 3 ) 



ph += 'g'


return (ph)

}

function D()

{


# D -> r | CA | FEGr


ph = ''


x = rand(1,3)


if ( x == 1 ) 



ph += 'r'


if ( x == 2 ) 



ph += 'c,a'


if ( x == 3 ) 



ph += 'f,e,g,r'


return (ph)

}

function E()

{


# E -> DG | A


ph = ''


x = rand(1,2)


if ( x == 1 ) 



ph += 'd,g'


if ( x == 2 ) 



ph += 'a'


return (ph)

}

function F()

{


# F -> A | B | Br


ph = ''


x = rand(1,3)


if ( x == 1 ) 



ph += 'a'


if ( x == 2 ) 



ph += 'b'


if ( x == 3 ) 



ph += 'b,r'


return (ph)

}

function G()

{


# G -> CF | A | E


ph = ''


x = rand(1,3)


if ( x == 1 ) 



ph += 'c,f'


if ( x == 2 ) 



ph += 'a'


if ( x == 3 ) 



ph += 'e'


return (ph)

}

function A()

{


# A -> Cr | BF


ph = ''


x = rand(1,2)


if ( x == 1 ) 



ph += 'c,r'


if ( x == 2 ) 



ph += 'b,f,r'


return (ph)

}

function B()

{


# B -> Dr | BG


ph = ''


x = rand(1,2)


if ( x == 1 ) 



ph += 'd,r'


if ( x == 2 ) 



ph += 'b,g'


return (ph)

}

function bgbeat(length)

{


ph = ''


for ( i=0; i< length; i+=192)


{



ph += 'p35d96, rd96'


}


ph.chan = 10


ph.vol = 50


return(ph)

}

function cut_phrase( ph, number ) {


# cuts a phrase to a number of notes


notes = 0


new_ph = ''


for ( note in ph ) {



if ( notes < number ) {




new_ph |= note



}



notes += 1


}


return (new_ph)

}

function pent(ph)

{


# uses the c major scale


# establish the scale


cpent = 'c1,d1,e1,g1,a1,c2,d2,e2,g2,a2,c3,d3,e3,g3,a3,c4,d4,e4,g4,a4,c5,d5,e5,g5,a5,c6,d6,e6,g6,a6,c7'


# create a 'blank' phrase


ph2 = ''


# for each note in the phrase, find closest note in the scale, and assign to the respective note in ph2


for ( nt in ph )


{



ph2 |= closest( nt,cpent )
# from default function library



}


return(ph2)

}

